

Using CRETH to make quantities add up without efficiency bias

Mark Horridge,

Centre of Policy Studies, Victoria University Melbourne

E-mail: mark.horridge@gmail.com

April 2019

Abstract

Modern CGE models can boast considerable sectoral detail. However, it is obvious that output of (say) electronic components, must be quite heterogeneous. Hence, since Leontief, multisectoral models tend to measure quantities not in physical units but in effective economic units (usually initial-dollars-worth).

The CET functional form, close cousin to CES, is used to allocate a fixed resource between alternate uses; for example land between crops, or workers between sectors. It works well when both input and output quantities are measured in initial-dollars-worth, such as land rental values. Because CET chooses a crop mix to maximize revenue, it is welfare-neutral -- a small change in land allocation will not affect land's contribution to GDP. This is a desirable property. But CET translates poorly into physical units: we typically find that if percent changes in (effective) land use are interpreted as percent changes in crop areas, then total land area is not fixed. This can be a problem for reporting results, or for interfacing a CGE model to ecological or agronomic models which work with physical units.

The CRETH functional form is a generalization of CET that has in the past been used like CET to allocate a fixed (measured in effective units) resource between alternate uses. In this usage, CRETH is like CET, but with more parameter flexibility. Here we show that CRETH land supply functions can instead be interpreted in a more literal fashion: as the answer (FOC) to a revenue-maximizing problem, where a land-owner allocates a fixed acreage of land between uses. Used in this way, CRETH (a) allows reported land areas to add up properly, and (b) has the optimum property that small changes in land allocation do not affect the land contribution to GDP (so avoiding efficiency bias).

1. Introduction

The CES functional form is heavily used in CGE modelling to combine several inputs into one aggregate. Cousin to CES is the less familiar CET form, which is used to split one thing into several. For example, CET might be used to

- split total output of some good into output for consumption and output for export.
- allocate a fixed stock of labour between industries.
- allocate a fixed stock of land between agricultural industries.

Below we focus on the last, land, example.

The CET equations often have the percentage change form:

$$x_i = x_{tot} + \sigma(p_i - p_{ave}) \quad \sigma > 0$$

$$p_{ave} = \sum_i S_i \cdot p_i \quad \text{where } S_i \text{ is the share of industry } i \text{ in total land rentals.}$$

which is the same as CES except for the positive sign on σ , the *Constant Elasticity of Transformation*. However, where high or infinite values of σ are regarded as plausible; it is necessary to rewrite the CET equations as:

$$(p_i - p_{ave}) = \tau(x_i - x_{tot}) \quad \text{where } \tau=1/\sigma \quad [\tau=0 \rightarrow \sigma=\infty]$$

$$x_{tot} = \sum_i S_i \cdot x_i$$

A simple TABLO implementation of the CET appears below:

CET1.TAB: a simple CET implementation

```

File INFILE;
! Data from file !
Set CROPS # Alternate land uses # read elements from file INFILE header "CROP";
Coefficient (all,c,CROPS) RENT(c) # Crop revenues #;
Read RENT from file INFILE header "RENT";
Coefficient (parameter) SIGMA # CET elasticity #;
Read SIGMA from file INFILE header "SIG";
! Derived coefficients !
Coefficient
TOTRENT # Total crop revenue #;
(all,c,CROPS) RENTSHR(c) # Revenue shares #;
Formula
TOTRENT = sum{c,CROPS, RENT(c)};
(all,c,CROPS) RENTSHR(c) = RENT(c)/TOTRENT;
Variable
(all,c,CROPS) p(c) # Rent per effective land units #;
(all,c,CROPS) x(c) # Quantities of effective land units #;
xtot # Total output (revenue-weighted) #;
pave # Average rent (revenue-weighted) #;
Update (all,c,CROPS) RENT(c) = p(c)*x(c);
Equation
E_pave pave = sum{c,CROPS, RENTSHR(c)*p(c)};
E_x (all,c,CROPS) x(c) = xtot + SIGMA*(p(c) - pave);

```

which is driven by the following CMF file¹:

CET1.CMF: simulation for CET1.TAB

```

auxiliary files = cet1;
File INFILE = DATA.HAR;
Updated File INFILE = <cmf>.UPD;
log file = yes;
method = Gragg ;
steps = 10 20 30;
Verbal Description = Wheat/Beef price changes;
Shock p("Wheat") = 20;
Shock p("Beef") = -20;
Exogenous p xtot;
Rest endogenous;

```

Data and results from CET1 appear in Table 1 below; results for p_{ave} and x_{tot} are shown in the Total row of the two final columns.

¹ Accompanying this document are all TAB, CMF, and HAR files needed to reproduce the simulations; see Appendix 1.

Table 1: Data and results from CET1.CMF

	Rent	RentShr	p	x
Wheat	5000	0.50	20.00	49.33
Fruit	3000	0.30	0.00	-39.99
Beef	2000	0.20	-20.00	-80.33
Total	10000	1.00	10.75	0.00

In DATA.HAR, the value of σ is 5. As we would expect, the output mix shifts strongly towards Wheat and away from Fruit and especially Beef.

So far, so good. The next step is to add to CET1.TAB some data for land areas (in hectares). As a check, we compute $qtot$, the change in total land area (we hope there is no change). Existing results are unaffected.

Additions for CET2.TAB: adding land areas

```
! As for CET1.TAB, plus .... !
Coefficient (all,c,CROPS) AREA(c) # Hectares used by crops #;
Read AREA from file INFILE header "AREA";
Coefficient
TOTAREA # Total crop area #;
(all,c,CROPS) AREASHR(c) # Area shares #;
Formula
TOTAREA = sum{c,CROPS, AREA(c)};
(all,c,CROPS) AREASHR(c) = AREA(c)/TOTAREA;
Update (all,c,CROPS) AREA(c) = x(c);
Variable qtot # Total land area #;
Equation E_qtot qtot = sum{c,CROPS, AREASHR(c)*x(c)};
```

Table 2: Data and simple CET results with land areas

	Rent	Area	RentShr	AreaShr	p	x	q
Wheat	5000	200	0.50	0.20	20.00	49.33	49.33
Fruit	3000	300	0.30	0.30	0.00	-39.99	-39.99
Beef	2000	500	0.20	0.50	-20.00	-80.33	-80.33
Total	10000	1000	1.00	1.00	10.75	0.00	-42.30

Above, we assume that $q(i)$, the %change in land area, follows $x(i)$, the "effective" land quantities, $x(i)$. Results for $qtot$ appear in the bottom right cell. We see quite a large change in total land area! This tells us that the $x(i)=q(i)$ cannot be used to report changes in land areas. The discrepancy between $xtot$ and $qtot$ is inevitable given the large differences between the $RentShr$ and the $AreaShr$ vectors. Such discrepancies might arise from 2 causes:

- Quite likely the $Rent$ and $Area$ values are drawn from different data sources which employ different conventions. For example, there may be differences in sectoral definitions. Perhaps more care with the data would reduce the problem.
- Even if the data are correct, the $RentShr$ and $AreaShr$ will diverge because average unit rent per hectare will differ between uses. In this example, the average unit rent per hectare is much lower for beef -- which is entirely plausible. We need to realize that for each crop there is a marginal rent per hectare which will differ from the average rent. At the margin of substitution between Beef land and Wheat land profits per hectare (marginal area rents) must be equal. That does not mean that average rents must be equal².

² Unfortunately the CET specification implies that the ratio of average to marginal rents is the same for all crops (see Section 4) -- which is part of the problem. That is not true for all other functional forms.

More generally the problem arises because the natural unit of measurement in CGE models is base-period-dollars-worth (the amount that initially may be bought with \$1). As soon as we introduce another unit of measurement (eg, hectares) problems may arise.

Although the units problem often appears in a CET or supply-side context, it may also appear in a CES or demand-side context. For example, an electricity distributor may purchase electricity from solar, coal or nuclear generator -- regarding these sources as imperfect substitutes. Average costs per kilowatt-hour will vary between sources -- implying that *qtot* and *xtot* measures of aggregate use will differ. Since, for the final user, a coal kilowatt-hour is indistinguishable from a solar kilowatt-hour, the difference is annoying to explain.

Again, results from a simulation where employment is fixed will depend on whether the fixed employment total is an hours-weighted or wage-weighted aggregate. Usually the latter is chosen, -- implying that one new employed dentist can compensate for 10 lost cleaning jobs.

As often in CGE, detail may come to our rescue. Suppose land were divided into numerous types, distinguished by climate, location and soil type. Then within one land type, we might expect that rent and area shares might be much closer, reducing the problems mentioned above. An example is the AEZ (Agro-Economic Zones) used in GTAP. Key here is that land is immobile: sandy soil in Texas never morphs into Illinois clay. Conversely, for labour supply we really do see a few dentists becoming farmers, so that occupational detail helps us less.

2. What if *qtot* were fixed?

A simple remedy might be to run CET2.TAB with *qtot* fixed (instead of *xtot*, see CET2A.CMF). With this closure the additional TABLO code for CET2.TAB *does* affect all results. The *x(i)* results are greatly changed. Indeed *qtot*, the total land area, is fixed; but *xtot*, the total economic contribution of land, varies quite a lot.

Table 2a: Data and results with fixed total land area

	RentShr	AreaShr	p	x	q
Wheat	0.50	0.20	20.00	158.79	158.79
Fruit	0.30	0.30	0.00	4.00	4.00
Beef	0.20	0.50	-20.00	-65.92	-65.92
Total	1.00	1.00	10.75	73.30	0.00

The large change in *xtot* raises severe difficulties with the welfare-oriented interpretation of results that is common in policy analysis. It implies that small changes in land allocation can affect GDP and other macro aggregates -- contrary to normal economic intuition. With *qtot* fixed, a hectare of beef land moving to wheat use enjoys an immediate sharp rise in per-hectare rents, causing aggregate output to rise. This is equivalent to treating the initial database as including a large distortion (good land wasted on the low-value Beef use). Hence, second-best effects will colour policy conclusions -- any policy is good which favours Wheat over Beef.

2.1. A related approach

The approach just described (*qtot* fixed) has no optimizing interpretation. The following idea addresses this problem. We assume that one land owner distributes land between uses according to the rule:³

³ Dixon and Rimmer (2003) suggested this approach for labour supply. Xin, Mensbrugghe and Tyner (2017) have dubbed it ACET and applied it to land supply. It is further discussed in Taheripour, Xin and Tyner (2018).

Choose X_i to maximize $U = \sum_i [P_i X_i]^\alpha$ such that $\sum_i X_i = Q_{tot}$
 [again we equate X_i with Q_i], giving rise to the following %change FOC:

$$x_i = q_{tot} + \sigma(p_i - p_{ave})$$

$$q_{tot} = \sum_i H_i \cdot x_i \quad \text{where } H_i \text{ is the share of industry } i \text{ in total land area.}$$

so here $p_{ave} = \sum_i H_i \cdot p_i$

Although ingenious, these FOC in fact generate the *same* results as CET2A.CMF.⁴ Hence we include no TAB file for this approach. All the criticisms of CET2A still apply. The non-optimal behaviour which was implicit in CET2A is here made explicit: the land owner does not value a dollar earned from Wheat as highly as a dollar earned from Beef: livestock (or at least variety) has special, non-monetary, attraction.

3. Methods of scaling land areas to add up

The basic framework of CET1.TAB uses data and variables that measure land in economic units [base-period-dollars-worth]. To tackle the problems described above, modellers have usually added to that framework a parallel system of data and variables which measures land in physical (hectare) units, as shown below (see also Figure 1 below).

Table 3: A parallel system of data and variables in physical units

	Economic units	Physical Units
Data	RENT(i), TOTRENT	AREA(i), TOTAREA
Shares	RENTSHR(i)	AREASHR(i)
Quantity variables	x(i), xtot	q(i), qtot
Price variables	p(i), ptot	r(i), rave

The added system enables reporting of results in physical units that add up correctly, yet it does not affect any of the results generated by CET1.TAB. Hence, substitution at the margin has no disturbing welfare effects.

It will be obvious from the preceding discussion that we cannot assume (as we did in CET2.TAB) that the percentage changes $x(i)$ and $q(i)$ are the same. For, if $x(i)=q(i)$, and $AREASHR(i) \neq RENTSHR(i)$, then $qtot \neq xtot$ (so they cannot both = 0, as usually desired).

Instead modellers assume, in levels, that

$$Q_i = F(X_i)$$

The choice of functional form F differs between practitioners (and is usually ad hoc).

3.1. Method 1

A common choice for F is:

$$Q_i = \Lambda X_i \quad \text{where } \Lambda \text{ is chosen so that } \sum_i Q_i = Q_{tot} \quad \text{with } Q_{tot} \text{ exogenous}$$

or, in percent change form:

⁴ Except that now $p_{ave} = \sum_i H_i \cdot p_i$

$$q_i = x_i + \lambda \quad \text{and} \quad \sum_i H_i q_i = q_{\text{tot}} \quad \text{where the } H_i \text{ are hectare shares.}$$

Such a system is shown in CET3.TAB below:

Additions for CET3.TAB: scaling land areas to add up, method 1

```
! As for CET1.TAB, plus .... !
Coefficient (all,c,CROPS) AREA(c) # Hectares used by crops #;
Read AREA from file INFILE header "AREA";
Coefficient
TOTAREA # Total crop area #;
(all,c,CROPS) AREASHR(c) # Area shares #;
Formula
TOTAREA = sum{c,CROPS, AREA(c)};
(all,c,CROPS) AREASHR(c) = AREA(c)/TOTAREA;
Variable (all,c,CROPS) q(c) # Percent change land areas #;
lambda # slack variable to allow correct land area addup #;
Update (all,c,CROPS) AREA(c) = q(c);
Equation E_q (all,c,CROPS) q(c) = x(c) + lambda;
Variable qtot # Total land area #;
Equation E_qtot qtot = sum{c,CROPS, AREASHR(c)*q(c)};
```

Results for the physical unit price variables, r_i and r_{ave} are not usually computed, but would be given by percent change equations⁵:

$$r_i + q_i = p_i + x_i \quad \text{and} \quad r_{\text{ave}} = \sum_i S_i r_i \quad \text{where the } S_i \text{ are revenue shares.}$$

New results using CET3.TAB are shown in the penultimate, q3, column of the table below. Note that results for p and x are the same as the original CET1 results. As desired, q_{tot} , the area-weighted average of the q_i , is 0. However, the q_i diverge quite widely from the x_i . In one case (Fruit) the sign is different! Should we report that land used for Fruit went up or down?

Table 4: Data and CET results with adjusted land areas

	RentShr	AreaShr	QRatio	p	x	q3	q4
Wheat	0.50	0.20	2.50	20.00	49.33	158.79	128.76
Fruit	0.30	0.30	1.00	0.00	-39.99	4.00	-38.18
Beef	0.20	0.50	0.40	-20.00	-80.33	-65.92	-28.60
Total	1.00	1.00	1.00	10.75	0.00	0.00	0.00

3.2. Method 2

Another choice for F is:

$$Q_i = X_i^{\alpha_i} \quad \text{where } \alpha_i = S_i/H_i \quad [= \text{RENTSHR}(i)/\text{AREASHR}(i)]$$

or, in percent change form:

$$q_i = \alpha_i x_i$$

Easy algebra will confirm that with this system we automatically obtain:

$$\sum_i S_i x_i = x_{\text{tot}} = \sum_i H_i q_i = q_{\text{tot}} \quad \text{ie, } q_{\text{tot}} \text{ need not be exogenous.}$$

Furthermore, x_i and q_i will always have the same sign.

Such a system is shown in CET4.TAB below:

⁵ These equations are in the supplied TAB files, but are not shown in the TAB excerpts presented in the text.

Additions for CET4.TAB: scaling land areas to add up, method 2

```
! As for CET1.TAB, plus .... !
Coefficient (all,c,CROPS) AREA(c) # Hectares used by crops #;
Read AREA from file INFILE header "AREA";
Coefficient
TOTAREA # Total crop area #;
(all,c,CROPS) AREASHR(c) # Area shares #;
Formula
TOTAREA = sum{c,CROPS, AREA(c)};
(all,c,CROPS) AREASHR(c) = AREA(c)/TOTAREA;
Coefficient (all,c,CROPS) QRATIO(c) # QRATIO=RENTSHR/AREASHR #;
Formula (all,c,CROPS) QRATIO(c) = RENTSHR(c)/AREASHR(c);

Variable (all,c,CROPS) q(c) # Percent change land areas #;
Update (all,c,CROPS) AREA(c) = q(c);
Equation E_q (all,c,CROPS) q(c) = QRATIO(c)*x(c);
Variable qtot # Total land area #;
Equation E_qtot qtot = sum{c,CROPS, AREASHR(c)*q(c)};
```

New results using CET4.TAB are shown in the last, q4, column of Table 4 above. As desired, q_{tot} , the area-weighted average of the q_i , is 0. It is not easy to say which of the q3 or q4 columns most nearly resemble the common x column. However, at least the q4 entries all have the same sign as their x counterparts.

Some theoretical support for the QRATIO formula is given by the next section, which proposes that a CRETH-like functional form may also lead to equations that allow quantities to add up properly, in both economic and quantity units.

4. Using CRETH for revenue maximisation

In this section we use a CRETH form to both conserve areas and maximize revenues.

CET, the supply analogue to CES, was introduced by Powell and Gruen (1968). CRESH, an elaboration of CES with more parameter flexibility, is due to Hanoch (1971). CRETH, supply analogue to CRESH, was introduced by Dixon, Vincent and Powell (1976).




Figure 1: Land distribution: economic and area units

We assume that a representative landowner, owning 1 unit of agricultural land⁶, rents this land to a number of tenant farmers, each growing a particular crop (indexed i). Some land suits particular crops, so that *effective land*, X_i , is a crop-specific function $F(Q_i)$ of the area used, Q_i . We assume that that farm output is, say, a CES aggregate of X_i and other inputs. Each farmer of type i , offers unit land rent P_i .

To maximize profit the landowner must choose Q_i (or X_i) to maximise revenue $R = \sum P_i X_i$ subject to the constraints $X_i = F(Q_i)$ and $\sum Q_i = 1$.

We observe initial $V_i = P_i X_i$ and the acreages Q_i . Revenue and acreage shares are denoted by:

$$S_i = V_i / \sum V_k \quad \text{and} \quad M_i = Q_i / \sum Q_k \quad [= Q_i]$$

The F function is: $X_i = F(Q_i) = A_i Q_i^{\alpha_i} \quad [0 < \alpha_i < 1]$

so $F'(Q_i) = [\alpha_i / Q_i] A_i Q_i^{\alpha_i} = [\alpha_i / Q_i] X_i$

The revenue maximizing problem is:

Choose Q_i to maximize $R = \sum P_i X_i = \sum P_i A_i Q_i^{\alpha_i}$ such that $\sum Q_i = 1$

Lagrangean = $\sum P_i A_i Q_i^{\alpha_i} - \Lambda [\sum Q_i - 1]$

Setting derivatives of Lagrangean w.r.t. Q_i to zero:

⁶ Assume if you like that the unit of land area is million hectares.

$$P_i[\alpha_i/Q_i]A_iQ_i^{\alpha_i} = P_iX_i[\alpha_i/Q_i] = \Lambda \quad \text{implying } \alpha_i \propto Q_i/S_i$$

or $\alpha_i = \alpha[Q_i/S_i]$ where α is a common value.

$$\sum Q_i = 1$$

We may observe that, if all the α_i were the same, so $\alpha_i = \alpha$, we would have the CET form. In that case the FOC $P_iX_i[\alpha_i/Q_i] = \Lambda$ would imply $P_iX_i \propto Q_i$, or that revenue and area shares are the same (ie, initial revenue per acre is equal for each crop). Actual data may not follow this rule.

Intuition: with the crop yield function: $X_i = A_iQ_i^{\alpha_i}$, the A_i are 'used up' in calibrating the S_i , so there is no flexibility left to accommodate different M_i .

Returning to the CRETH system we get the percentage change forms:

$$p_i + (\alpha_i - 1)q_i = \lambda \quad \text{or} \quad q_i = [\lambda - p_i]/(\alpha_i - 1) \quad q_i = \sigma_i p_i - \sigma_i \lambda \quad \text{where } \sigma_i = 1/(1 - \alpha_i)$$

$$\text{or} \quad q_i = \sigma_i [p_i - \lambda]$$

$$\sum M_i q_i = 0 \quad \text{where } M_i = Q_i / \sum_k Q_k$$

$$x_i = \alpha_i q_i$$

Now use constraint $\sum M_i q_i = 0$ to eliminate λ :

$$\sum M_i q_i = 0 = \sum M_i \sigma_i [p_i - \lambda] = 0$$

$$\text{so} \quad \lambda = \sum B_k p_k \quad \text{where} \quad B_k = M_k \sigma_k / \sum_i M_i \sigma_i \quad (\text{the modified CRETH shares})$$

In summary, our percentage change equations are:

$$q_i = \sigma_i [p_i - \lambda]$$

$$\lambda = \sum B_k p_k \quad \text{or} \quad \sum M_i q_i = 0 \quad \text{where } M_i = Q_i / \sum_k Q_k$$

$$x_i = \alpha_i q_i$$

and for initial calibration, compute

$$S_i = V_i / \sum V_k \quad \text{and} \quad M_i = Q_i / \sum Q_k$$

$$\alpha_i = \alpha [M_i / S_i] \quad \text{where } \alpha \text{ is a common value.}$$

$$\sigma_i = 1/(1 - \alpha_i)$$

Lastly, recall that our equations apply to a representative landlord with one unit of land. Then a 1% increase in total land supply is equivalent to a 1% increase in the number of landlords. Let Q = the total land area (or no. of landlords) and use bold symbols \mathbf{q}_i and \mathbf{x}_i to denote percentage changes in total supplies. Hence:

$$\mathbf{q}_i = q + q_i \quad \text{or} \quad q_i = \mathbf{q}_i - q$$

$$\text{and} \quad \mathbf{x}_i = q + x_i \quad \text{or} \quad x_i = \mathbf{x}_i - q$$

Then our total supply equations are:

$$\mathbf{q}_i = q + \sigma_i [p_i - \lambda]$$

$$\lambda = \sum B_k p_k \quad \text{or} \quad \sum M_i (\mathbf{q}_i - q) = 0 \quad \text{where } M_i = Q_i / Q$$

$$\mathbf{x}_i - q = \alpha_i (\mathbf{q}_i - q)$$

The above forms are those which appear in the TABLO file CRETH.TAB.

Setting $\alpha = 0.3$, and repeating our previous experiment, we obtain Table 5 below. Note the uniformity of the final column, which arises from our assumption that at the margin, a hectare of land earns the same, however used.

Table 5: Some CRETH results

	RentShr	AreaShr	Alpha	Sigma	p	x	q	r
Wheat	0.5	0.2	0.12	1.14	20	4.4	43.3	-12.5
Fruit	0.3	0.3	0.30	1.43	0	5.9	21.1	-12.5
Beef	0.2	0.5	0.75	4.00	-20	-23.4	-30.0	-12.5
Total	1	1	0.30		6.7	0	0	-12.5

Appendix 2 attempts a graphical motivation for the CRETH supply functions.

5. General reasons to prefer a revenue-maximization approach

By default, CGE models have traditionally assumed that agents are profit-maximizers or cost-minimizers. Reasons to follow this tradition include:

- Optimizing assumptions impose restrictions on response elasticities, helping to reduce uncertain parameter space.
- We may really believe that, in the mass, agents act as if only profit-seeking. Certainly one farmer may prefer to grow steers even if growing brussel sprouts might earn more. But bank managers and old age will constantly reduce the number of such farmers.
- Non-optimizing behaviour may be thought of a distortion, which can bias policy prescriptions. As explained in Section 2 above, policies which penalize beef-eaters may, for second-best reasons, increase GDP. Perhaps indeed people should indeed eat more fruit and less beef! But we need to make an argument or bring evidence to support this conclusion -- it should not merely emerge from an assumption about land supply.

The CRETH form suggested above is the simplest mechanism which can be reconciled (or calibrated) with observed land rentals and acreages. It is a natural starting point for more elaborate modelling.

6. Conclusion

We reviewed the CET functional form, and explained the difficulty that arises in getting physical quantities to add up correctly. Several possible solutions were reviewed -- none were completely satisfactory. We found that a CRETH supply system could be both area-preserving and revenue-maximizing.

7. References

- Peter B. Dixon & David P. Vincent & Alan A. Powell (1976), "Factor Demand and Product Supply Relations in Australian Agriculture : The CRESH/CRETH Production System," Centre of Policy Studies/IMPACT Centre Working Papers op-08, Centre of Policy Studies, Victoria University,
- Dixon, P.B., and M.T. Rimmer (2003), "A New Specification of Labour Supply in the MONASH Model with an Illustrative Application." *Australian Economic Review* 36:22–40. doi:10.1111/1467-8462.00265.
- Hanoch, Giora (1971), "CRESH Production Functions," *Econometrica*, Econometric Society, vol. 39(5), pages 695-712, September.
- Alan A. Powell and F. H. G. Gruen (1968), "The Constant Elasticity of Transformation Production Frontier and Linear Supply System", *International Economic Review*, Vol. 9, No. 3 (Oct., 1968), pp. 315-328
- Taheripour, Farzad, Xin Zhao and Wally Tyner (2018), "Modeling land use in large scale global computable general equilibrium models: Preserving physical area of land", paper presented at the 21st Annual Conference on Global Economic Analysis, Cartagena, Colombia

Zhao, Xin, Dominique van der Mensbrugge and Wally Tyner (2017), "Modeling land physically in CGE models: new insights on intensive and extensive margins", paper presented at the 20th Annual Conference on Global Economic Analysis, Purdue University, USA

Appendix 1: Files distributed with this document


A zip archive accompanies this document; it contains:

- TAB and CMF files so you can run all the examples mentioned in the text. Type "RunSims.bat" from the command line to run all the examples.
- A complete set of SL4 solution files which you can examine even if you do not want to run all the simulations.

See Archive item TPMH0181 at <http://www.copsmodels.com/archivep.htm>

Appendix 2: Why is high rent land less substitutable between uses ?

We attempt to graphically motivate the CRETH distribution system. In the diagram below, a fixed acreage of land has been arranged along a horizontal spectrum, so that the more fertile acres are grouped at the left, the remainder at the right. The land can be used for crops or sheep; crop acreage is measured from the left; sheep from the right. Two curves indicate the marginal value of an additional acre allocated to each use. We see that for crops marginal value declines fairly steeply as the best land get used up; but decline much less steeply for sheep (where any land will do).



We measure both crop and sheep land earnings in dollars-worth (ie, choose units so that output prices are 1). The margin of cultivation is located at the intersection of the two marginal [value of] output curves. We choose yield functions consistent with the diagram:

$$\text{Crop Earnings} = B_c L_c^{0.5}$$

$$\text{Value of additional crop acre} = B_c L_c^{-0.5} / 2$$

$$\text{Rent per Acre} = B_c L_c^{-0.5}$$

$$\text{Sheep earning} = B_s L_s^{0.999}$$

$$\text{Value of additional sheep acre} = B_s$$

$$\text{Rent per Acre} = B_s$$

$$\text{Rental ratio} = R = \text{rent per crop acre} / \text{rent per sheep acre} = [B_c / B_s] L_c^{-0.5}$$

$$\text{FOC: } B_c L_c^{-0.5} / 2 = B_s$$

$$\text{or } [B_c / B_s] L_c^{-0.5} = 2$$

$$\text{so } R = 2$$

Intuition: the steeply declining marginal product curve means tight curvature of the crop yield function, leading to low substitutability. But steeply declining marginal product also implies that infra-marginal and hence average yields are much greater than marginal yields, leading to high average rentals.

CRETH assumes (a) smoothly declining marginal yields; and (b) no special pair-wise relations of substitutability. Interesting thought-experiments, which seem to challenge the rent- σ relation, often rely on breaking (a) or (b).